

Integration von leJOS in die Eclipse IDE (Helios)

Aegidius Plüss, Version 1.0, 23. September 2011

Voraussetzungen (als Beispiel, die Verzeichnisse können anders lauten):

- Windows mit Pfad auf `c:\bat`
- leJOS im Verzeichnis `e:Vejos`, Pfad auf `e:Vejos\bin`, Umgebungsvariable `NXJ_HOME` auf `e:Vejos`

Erstellen (oder Herunterladen) von 3 Batch-Dateien:

lejoslink.bat:

```
@echo off

rem Version 1.0 (www.aplu.ch)
rem Calls nxjlink.bat as Eclipse external tool
rem Assumptions:
rem - NXJ_HOME environment variable defined
rem - lejos_home\bin part of system path

if .%2 == . goto usage
if not .%3 == . goto usage

echo leJOS linking now...

set home=%1
set file=%2
echo Project home = %home%
echo Selected Java source file = %file%

rem Strip file extension
set classname=%file:~0,-5%
echo Java class name = %classname%

rem Call leJOS nxjlink from directory of .class file
cd /d %1\bin
call nxjlink %classname% -o%classname%.nxj

echo Linking finished.
goto quit

:usage
echo Usage: lejoslink ^<projectpath^> ^<sourcefile^>.java

:quit
```

lejosupload.bat:

```
@echo off

rem lejosupload.bat
```

```

rem Version 1.0 (www.aplu.ch)
rem Calls nxjupload.bat as Eclipse external tool
rem Assumptions:
rem - NXJ_HOME environment variable defined
rem - lejos_home\bin part of system path

if .%2 == . goto usage
if not .%4 == . goto usage

echo leJOS uploading now...

set home=%1
set file=%2
set bluetoothname=%3

echo Project home = %home%
echo Selected Java source file = %file%
echo Using Bluetooth name = %bluetoothname%

rem Strip file extension
set classname=%file:~0,-5%
echo Java class name = %classname%

rem Call leJOS nxjupload from directory of .class file
cd /d %1\bin
call nxjupload -r -b -n %bluetoothname% %classname%.nxj

rem Remove -b and -n parameter if USB upload is needed
rem call nxjupload -r %classname%.nxj

echo Uploading finished.
goto quit

:usage
echo Usage: lejosupload ^<projectpath^> ^<sourcefile^>.java ^<bluetoothname^>

:quit

```

lejoslinkandupload.bat:

```

@echo off

rem lejoslinkandupload.bat
rem Version 1.0 (www.aplu.ch)
rem Calls nxjlink.bat and nxjupload.bat as Eclipse external tool
rem Assumptions:
rem - NXJ_HOME environment variable defined
rem - lejos_home\bin part of system path

if .%2 == . goto usage
if not .%4 == . goto usage

```

```

call lejoslink.bat %1 %2
call lejosupload.bat %1 %2 %3
goto quit

:usage
  echo Usage: lejoslinkandupload ^<projectpath^> ^<sourcefile^>.java ^<blue
toothname^>

:quit

```

In *lejosupload.bat* die Zeile mit *nxjupload* den eigenen Bedürfnissen anpassen (NXT Bluetooth Name ändern, event. den -b und -n Parameter weglassen, wenn USB verwendet wird).

Wie man sieht, erwarten diese Batch-Dateien 3 Parameter: das Homeverzeichnis des Projekts, den Dateinamen inklusive .java der Java-Quelldatei und den NXT Bluetooth Namen (falls gebraucht). Diese Parameter kann man mit Eclipse automatisch generieren lassen, wenn man ein **External Tool** erstellt.

Vorgehen:

1. Eclipse starten
2. Menu *Run | External Tools | External Tools Configuration*
3. Im linken Fenster rechter Mausklick auf *Program*. Dann *New* wählen
4. Im rechten Fenster

Name: *LejosLinkAndUpload* (oder anders)

Location: *c:\bat\lejoslinkandupload.bat*

Arguments:

```

${project_loc} ${selected_resource_name} NXT

```

eintippen, wo NXT der

verwendete NXT Bluetooth Name ist.

4. *Apply* klicken (Run führt zu einer Fehlermeldung, da *selected_resource_name* nicht definiert ist)
5. Reiter *Common* klicken. Unter *Display in favorites menu* die Checkbox *External Tools* selektieren. *Apply* klicken

Nun kann ein Eclipse Project wie ein gewöhnliches Java-Projekt erstellt werden. Als External Jars muss die leJOS library *classes.jar* ausgewählt werden. Falls die Bibliothek *NxtJLibA* von <http://www.aplu.ch/nxt> verwendet werden soll, muss das Original, das sich in einem Unterverzeichnis des lejos-home-Verzeichnis befindet, durch die in der NxtJLibA-Distribution enthaltene Library *classes.jar* **ersetzt** werden (Backup vom Original erstellen).

Im Package Explorer die Quelldatei durch Mausklick selektieren. Speichern (*Ctrl-S*) kompiliert die Datei (die Classdatei befindet sich im Projektordner *bin*). Nun unter Menu *Run | External Tool* das angezeigte Tool *LejosLinkAndUpload* anklicken und es sollte im Console Fenster die korrekte Abarbeitung sichtbar sein. **Achtung: es muss immer im Package Explorer die Quelldatei aktiv selektiert sein, wenn das External Tool gestartet wird!**

Man kann die Bedienung weiter vereinfachen, indem man einen **Shortcut** für die Ausführung des *Last External Tools* definiert. Wie folgt vorgehen:

Menu *Window | Preferences*. Im linken Fenster *General* öffnen und *Keys* klicken. Im Feld *Filter external* eintippen und *Command Run Last Launched External Tool* wählen.

Feld *Binding*: wählen und auf der Tastatur einen Shortcut eintippen, z.B. *Ctrl-T*

Ergänzung:

Man kann die **Quick Access** Option von Eclipse verwenden, um eine bestimmte Option mit einem Tastendruck auszuführen. Dazu drückt man *Ctrl-3* und schreibt dann auf der Tastatur *external*. Aus dem erscheinenden Menu wählt man *External Tools - Run LejosLinkAndUpload* aus. Von jetzt an kann man *Ctrl-3* drücken, um *LejosLinkAndUpload.bat* auszuführen. Hat man gerade im Editor gearbeitet, so genügt dann die Tastenfolge **Ctrl-3, Enter, Enter** um das Programm auf dem NXT auszuführen. Alle Ausgaben der Batch-Dateien sind im Eclipse Console-Window sichtbar. Es sind auch mehrere Quelldateien im selben Projekt erlaubt. Man markiert im Projektmanager einfach diejenige, die verwendet werden soll.

Hinweise:

Für Tests können die Batch-Dateien auch direkt unter Verwendung entsprechender Parameter aufgerufen werden.

Download der Batch-Dateien von <http://www.aplu.ch/home/download/eclipsebatch.zip>

Kommentare an support@aplu.ch sind willkommen!